

NAG Library Routine Document

G05RGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G05RGF generates pseudorandom uniform bivariates with joint distribution of a Plackett copula.

2 Specification

```
SUBROUTINE G05RGF (N, THETA, SORDER, STATE, X, LDX, SDX, IFAIL)
  INTEGER          N, SORDER, STATE(*), LDX, SDX, IFAIL
  REAL (KIND=nag_wp) THETA, X(LDX,SDX)
```

3 Description

Generates pseudorandom uniform bivariates $\{u_1, u_2\} \in [0, 1]^2$ whose joint distribution is the Plackett copula C_θ with parameter θ , given by

$$C_\theta = \frac{[1 + (\theta - 1)(u_1 + u_2)] - \sqrt{[1 + (\theta - 1)(u_1 + u_2)]^2 - 4u_1u_2\theta(\theta - 1)}}{2(\theta - 1)}, \quad \theta \in (0, \infty) \setminus \{1\}$$

with the special cases:

$C_0 = \max(u_1 + u_2 - 1, 0)$, the Fréchet–Hoeffding lower bound;

$C_1 = u_1u_2$, the product copula;

$C_\infty = \min(u_1, u_2)$, the Fréchet–Hoeffding upper bound.

The generation method uses conditional sampling.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05RGF.

4 References

Nelsen R B (2006) *An Introduction to Copulas* (2nd Edition) Springer Series in Statistics

5 Parameters

- | | | |
|----|--|--------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of bivariates to generate. | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 2: | THETA – REAL (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> θ , the copula parameter. | |
| | <i>Constraint:</i> THETA ≥ 0.0 . | |

- 3: SORDER – INTEGER *Input*
On entry: determines the storage order of variates; the (i, j) th variate is stored in $X(i, j)$ if $SORDER = 1$, and $X(j, i)$ if $SORDER = 2$, for $i = 1, 2, \dots, n$ and $j = 1, 2$.
Constraint: $SORDER = 1$ or 2 .
- 4: STATE(*) – INTEGER array *Communication Array*
Note: the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 5: X(LDX, SDX) – REAL (KIND=nag_wp) array *Output*
On exit: the n bivariate uniforms with joint distribution described by C_θ , with $X(i, j)$ holding the i th value for the j th dimension if $SORDER = 1$ and the j th value for the i th dimension if $SORDER = 2$.
- 6: LDX – INTEGER *Input*
On entry: the first dimension of the array X as declared in the (sub)program from which G05RGF is called.
Constraints:
 if $SORDER = 1$, $LDX \geq N$;
 if $SORDER = 2$, $LDX \geq 2$.
- 7: SDX – INTEGER *Input*
On entry: the second dimension of the array X as declared in the (sub)program from which G05RGF is called.
Constraints:
 if $SORDER = 1$, $SDX \geq 2$;
 if $SORDER = 2$, $SDX \geq N$.
- 8: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, corrupt STATE parameter.

IFAIL = 2

On entry, invalid THETA: THETA = $\langle value \rangle$.
Constraint: THETA \geq 0.0.

IFAIL = 3

On entry, N = $\langle value \rangle$.
Constraint: N \geq 0.

IFAIL = 4

On entry, invalid SORDER.
Constraint: SORDER = 1 or 2.

IFAIL = 6

On entry, LDX must be at least $\langle value \rangle$: LDX = $\langle value \rangle$.

IFAIL = 7

On entry, SDX must be at least $\langle value \rangle$: SDX = $\langle value \rangle$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.
See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.
See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.
See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G05RGF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

In practice, the need for numerical stability restricts the range of θ such that:

if $\theta < \epsilon_s$, the routine returns pseudorandom uniform variates with C_0 joint distribution;

if $|\theta - 1| < \epsilon$, the routine returns pseudorandom uniform variates with C_1 joint distribution;

if $\theta > \epsilon_s^{-1/2}$, the routine returns pseudorandom uniform variates with C_∞ joint distribution;

where ϵ_s is the safe-range parameter, the value of which is returned by X02AMF; and ϵ is the *machine precision* returned by X02AJF.

10 Example

This example generates thirteen variates for copula $C_{2,0}$.

10.1 Program Text

```

Program g05rgfe

!      G05RGF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: g05kff, g05rgf, nag_wp, x04caf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: lseed = 1, nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: theta
Integer                    :: genid, ifail, ldx, lstate, n, sdx,    &
                          sorder, subid
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: x(:, :)
Integer                    :: seed(lseed)
Integer, Allocatable        :: state(:)
!      .. Executable Statements ..
Write (nout,*) 'G05RGF Example Program Results'
Write (nout,*)
Flush (nout)

!      Skip heading in data file
Read (nin,*)

!      Read in the base generator information and seed
Read (nin,*) genid, subid, seed(1)

!      Initial call to initialiser to get size of STATE array
lstate = 0
Allocate (state(lstate))
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Reallocate STATE
Deallocate (state)
Allocate (state(lstate))

!      Initialize the generator to a repeatable sequence
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Read in sample size and order
Read (nin,*) n, sorder

      If (sorder==1) Then
!          X(N,2)
          ldx = n
          sdx = 2
      Else
!          X(2,N)
          ldx = 2
          sdx = n
      End If
      Allocate (x(ldx,sdx))

!      Read in parameter
Read (nin,*) theta

!      Generate variates
ifail = 0

```

```

      Call g05rgf(n,theta,sorder,state,x,ldx,sdx,ifail)

!      Display the variates
      If (sorder==1) Then
!        X(N,2)
          ifail = 0
          Call x04caf('General',' ',n,2,x,ldx, &
            'Uniform variates with copula joint distribution',ifail)
      Else
!        X(2,N)
          ifail = 0
          Call x04caf('General',' ',2,n,x,ldx, &
            'Uniform variates with copula joint distribution',ifail)
      End If

      End Program g05rgfe

```

10.2 Program Data

```

G05RGF Example Program Data
1 1 1762543      :: GENID,SUBID,SEED(1)
13 1           :: N,SORDER
2.0           :: THETA

```

10.3 Program Results

G05RGF Example Program Results

Uniform variates with copula joint distribution

	1	2
1	0.6364	0.0695
2	0.1065	0.4586
3	0.7460	0.3586
4	0.7983	0.3267
5	0.1046	0.9888
6	0.4925	0.8920
7	0.3843	0.4903
8	0.7871	0.4248
9	0.4982	0.5783
10	0.6717	0.4419
11	0.0505	0.5802
12	0.2580	0.6629
13	0.6238	0.3291
